```
DEFINE_CALL TOTASS(INPUT,INPUT,INPUT,INPUT,INPUT, ...
                   INPUT, INPUT,INPUT,INPUT,      ...
                   OUTPUT,OUTPUT,OUTPUT,OUTPUT,OUTPUT)

TITLE Physiological process-based model for ....
      growth and development of cotton: Potential yield

*************************************************************************
*                              SUCROS-Cotton                            *
*           A simple Physiological based simulation model for           *
*                 potential cotton growth and development               *
*                                                                       *
*                                                                       *
*                          (FST-version 2.0)                            *
*                            Lizhen Zhang                               *
*                     China Agricultural University                     *
*                            Beijing,  P.R.China                        *
*                     Email:Zhang.lizhen@hotmail.com                    *
*                                                                       *
* *********************************************************************


ARRAY BOLL(1:N)  ,BOLLI(1:N)  ,RBOLL(1:N) ,CBOLL(1:N),NETFLO(1:N),...
      FLOW(1:N+1),AGE(1:N)    ,WBAGE(1:N) ,RWB(1:N)  ,...
      STRESS(1:N),RWBAGE(1:N) ,WBOLL(1:N)            ,...
      CWBOLL(1:N),WFLOW(1:N+1),WNETFL(1:N),RWBOL(1:N),...
      SHELL(1:N) ,SEDCOT(1:N)
* WBOLLP(1:N),

INITIAL

INCON ZERO =0.0
INCON IW   =7.8   ;IWLV=4.2;IROOT=1.8;ISTEM=1.8;ISHOT=5.5661
INCON ILAI =0.006 ;IWBOLL=1.92
INCON BOLLI=0.    ;BOLL0I=0.
PARAM GF   =70.   ;PLANTS=60000.
ARRAY_SIZE N=74

WEATHER WTRDIR='E:\WEATHER\ANYANG\';CNTR='AY';ISTN=1;IYEAR=2001

* Reading weather data:
* RDD   Daily global radiation       J/m2/d
* TMMN  Daily minimum temperature    oC
* TMMX  Daily maximum temperature    oC
* VP    Vapour pressure              kPa
* WN    Wind speed                   m/s
* RAIN  Precipitation                mm
* LAT   Latitude of the site         degree

TIMER STTIME =113.; FINTIM =360.; PRDEL =1. ; DELT = 1.

PRINT PDT,LAI,BIOMUP,WSHOOT,TDRW,WLEAF,WSTEM,BOLLW,ROOT,ROOTA,...
      HEIGHT,LEAFM,FBA,FN,SQUARE,FLOWER,SBOLL,BBOLL,BOLOUT,GREBL,...
      COTTON,COTOUT,COTGRE,LINT,LINTO,LINTG,RLINT,LINTM,LINTT,...
      WSOUT,COTR,STRBOL,PISHOT,PIROOT,FLT,FALL1,FALL2,FALL3,...
      GLAI,GTW,RWBT,RWBL,RWLV,RWSM,FALLBL,RTE,...
      RPBO,POOL,RGRM,RGUP,MAINT

DYNAMIC

* 1.  Physiological development
```

```
      RPDT=MAX(0., RTE*RPE*VI)
      PDT =INTGRL(ZERO, RPDT)
* PDT is physiological development time, including thermal effectiveness
* and photoperiod effectiveness, variety maturity effect

* 1.1 Thermal effectiveness,RTE

*   Temperature table
*   Temperature at      TBase  TOptimal TMax
*     Emergence          15     25-30    40   (soil temperature)
*     Square             12     30-35    45
*     Flowering          12     30-35    45
*     Boll opening       10     26-30    35
* variables:
* RTE   relative thermal effectiveness
* RTEB  temp. table for air temperature
* RTESB temp. table for soil temperature
* TSAV  soil temperature in degree c
* TSCAV soil temperature when film mulched in degree c
* TAV   average daily air temperature in degree c
* TPLUS intermediate variable for temp. compensation when film mulched
* IE    compensation coeffient before square is 0.51, flower 0.22,
*       after flower 0, quote from experiments China
*       assump daily temperature are 50% of average and 25% of maximum
*       and 25% of minimum

FUNCTION RTETB = -20.,0., 12.,0., 30.,1.0, 35.,0., 45.,0.
FUNCTION RTESTB= -20.,0., 15.,0., 20.,0.3, 30.,1., 35., 0., 45.,0.

     TAV =(TMMN+TMMX)/2.
     TSAV=0.890 + 1.017*TAV
* soil temperature will be used in simulation of emergence

* effect of film mulching FILM
* 0.: without film mulching; 1.: with film mulching
PARAM FILM=0.
     TSCAV=7.5725 + 0.8303*TAV
     TPLUS=MAX(0.,((TSCAV-TSAV)/NOTNUL((TSCAV-12.)/NOTNUL(TAV-12.)))*IE)
     IE   =INSW(PDT-17.5, 0.51, TETMP)
     TETMP=INSW(PDT-27.5, 0.22, 0.)

* RTE calculations
     RFEE  =AFGEN(RTESTB,TSCAV)
     RTEE1 =AFGEN(RTESTB,TSAV )
     RTEE  =INSW(FILM-1.,RTEE1,RFEE)
     TAV1  =INSW(FILM-1.,TAV  ,TAV+TPLUS)
     TMMN1 =INSW(FILM-1.,TMMN ,TMMN+TPLUS)
     TMMX1 =INSW(FILM-1.,TMMX ,TMMX+TPLUS)
     RTEFSB=0.25*( 2.*AFGEN(RTETB,TAV1) + AFGEN(RTETB,TMMN1) + ...
                                          AFGEN(RTETB,TMMX1) )
     RTE   =INSW(PDT-2.5, RTEE, RTEFSB)

* 1.2 Photoperiod effectiveness, RPE
*   RPE   relative photoperiod effectiveness due to daylength
*         it effect of flowering time, in the most regions this effect
*         can be ignored for upland cotton cultivars

     RPE=AFGEN(DLTB,DAYL)
FUNCTION DLTB=8.0,1.,12.,1.,14.,1., 16.,0.

* 1.3 Variety index, VI
```

```
* VI be vauled according to the maturing factor
* early matured cotton cultivars,        0.96-1.00,  such as CRI37
* early middle matured cotton cultivars, 0.82-0.84,  such as CRI32,41
* middle matured cotton cultivars,       0.80-0.81,  such as CRI12
* this parameter can be obtained from experiments or generally for above

PARAM VI=0.83


* 2. Cotton morphogenesis

* cutoff practice
* normally, farmer cut off top of cotton plant to stop growth of main
* stem in oder to avoid unuseful square and bolls,time of cutoff is
* around 12-14 fruit branches at density of 60000 per ha, more
* population less fruit branch needed, cutoff time will be earlier.
* day 208, is the date about 30 July in China.
* FFB  variety factor, first friut branch lacated in the main stem

PARAM FFB=9.
PARAM CUT=208.

   CUTOFF=INSW(TIME-CUT, 0., 1.)

* 2.1 Plant height
* HEIGHT  plant height                    cm
* RHR     rate of height growth           cm.d-1
* FLT     temperature factor

     RHR1  =1.5*FLT
     RHR2  =INSW(PDT-2.5  ,0.  ,RHR1)
     RHR   =INSW(CUTOFF-1.,RHR2,0.)
     HEIGHT=INTGRL(ZERO, RHR)

* 2.2 Leaf number in main stem
* RLR      rate of new leaf development       leaf.d-1
* LEAFA    actual leaf number                 leaf
* RLDR     leaf number abscision rate         leaf.d-1
* LEAFD    differentied leaf number           leaf
* LEAFP    potential total leaf number        leaf
* LEAFM    all happened leaf number           leaf

     RLR1 =0.5998*RTE-0.0594
     RLR2 =INSW(PDT-2.5,0.,MAX(0.,RLR1))
     RLR3 =INSW(CUTOFF-1.,RLR2,0.)
     RLDR1=1./70.*LEAFA
     RLR  =RLR3-RLDR1
     LEAFM=INTGRL(ZERO,RLR3)
     LEAFA=INTGRL(ZERO,RLR)
     LEAFD=AFGEN(LDTB,LEAFM)
*    LEAFP=LEAFM+LEAFD

FUNCTION LDTB= 0.,4., 1.,5., 5.,9., 11.,9., 12.,10., 30.,11., ...
              50.,15.

* 2.3 Fruit branch
* RFBR     rate of new fruit branch development     number.d-1
* FBA      actual fruit branch calculated by rate
* FBAV     actual fruit branch calculated by leaf number
* FBAD     differentied fruit branch
```

```
* FBAP     potential fruit branch total

    RFBR1=1.5459*RTE-0.7712
    RFBR2=INSW(LEAFM-FFB,0.,MAX(RLR3,RFBR1))
    RFBR =INSW(CUTOFF-1.,RFBR2,0.)
    FBA  =INTGRL(ZERO,RFBR)

* for check fruit branch number
    FBAV1=LEAFM-FFB
    FBAV =INSW(LEAFM-FFB, 0., FBAV1)
    FBAD =INSW(LEAFM-FFB, 0., LEAFD-2.)
    FBAP =FBAV+FBAD

* 2.4 Fruit node
* RFNR     rate of new fruit node development      number.d-1
* FN       actual fruit node
* FNAVM    potential fruit node after cutoff

    RFNR1=5.8746*RTE-3.2508
    RFNR2=INSW(LEAFM-FFB,0.,MAX(RFBR,RFNR1))
    RFNR =INSW(FN-FNAVM,RFNR2,0.)
    FNAVM=AFGEN(FNTB,FBAP)
FUNCTION FNTB=0., 0., ...
             1., 1., 2., 2., 3., 3., 4., 5., 5., 7., 6., 9., 7.,12.,...
             8.,15., 9.,18.,10.,22.,11.,26.,12.,30.,13.,35.,14.,40.,...
            15.,45.,16.,51.,17.,57.,18.,63.,19.,70.,20.,77.,21.,84.,...
            22.,92., 23.,100., 24.,108., 25.,117., 26.,126., ...
            36.,234., 60., 500.

    FN=INTGRL(ZERO,RFNR)

* 3. LAI calculation
* GLAI    growth rate of LAI                    lai.d-1
* FLV     partitioning index of leaf
* RWLV    growth rate of leaf weight            kg.hm-2.d-1
* RGUP    dry matter for daily shoot growth     kg.hm-2.d-1
* RLDR    LAI abcision rate                     lai.d-1
* SLA     the ratio of leaf eare and weight     hm2/kg

   GLAI=INSW(PDT-2.5, 0., RWLV*SLA*FLT - RLDR*LAI)
   RLDR=AFGEN(RLDRTB,PDT)
   SLA =AFGEN(SLATB ,PDT)
   LAI =INTGRL(ILAI ,GLAI)
FUNCTION SLATB =0.,0.,  2.5,0.00164, 27.5,0.0022, 80.,0.00136, ...
                    100.,0.00136
FUNCTION RLDRTB=0.,0., 17.5,0.0,     27.5,0.0, 60.,0.03, ...
                    100., 0.06

*   Temperature modification
    FLT1=1. - 0.003*(TAV-30.)**2.
    FLT2=MAX(0.  ,FLT1)
    FLT =MIN(FLT2,1.  )
*   FLT  temperature effect on growth rate


* 4. Dry matter partitioning

* 4.1 Long-term pool
*  Assumption of existence of long-term pool, that maintain cotton
*  growth when daily photothesis rate are low
*  SCR     daily used dry matter from long term pool    kg.hm-2.d-1
```

```
*  RGCHO   dry matter used for daily growth             kg.hm-2.d-1
*  RPUP    rate of dry matter growth of long term pool  kg.hm-2.d-1
*  GTW     rate of dry matter growth total              kg.hm-2.d-1
*  RPBO    daily drymatter left when supply bigger than demand of boll
*          also pile up in the long-term pool
*  RWBL    dry matter supply to boll growth             kg.hm-2.d-1
*  RWBT    dry matter demand of boll growth             kg.hm-2.d-1
*  POOL    long-term pool                              kg.hm-2

  SCR   =INSW(GTW-0.,0.2*POOL,0.1*POOL)
  GROWTH=0.7*GTW+SCR
  RPUP  =0.3*GTW-SCR+RPBO
  RPBO  =MAX(0.,(RWBL-RWBT))
  POOL  =INTGRL(ZERO,RPUP)

* 4.2 Partitioning to organs

* 4.2.1 Root
*  RGRA  actual growth rate of root                 kg.hm-2.d-1
*  RGRM  maximum growth rate of root                kg.hm-2.d-1
*  ROOT  total dry matter partitioning to root      kg.hm-2
*  ROOTA actual root weight                         kg.hm-2

   RGRM  =AFGEN(ROOTTB,PDT)*GTW
   RDROOT=AFGEN(RDTB,PDT)
   ROOT  =INTGRL(IROOT,RGRM)
   RGRA  =RGRM-RDROOT*ROOTA
   ROOTA =INTGRL(IROOT,RGRA)
FUNCTION ROOTTB=0.,0.33, 2.5,0.33, 17.5,0.33, 27.5,0.0, 100.,0.
FUNCTION RDTB  =0.,0.,   2.5,0.,   17.5,0.,   70.,0.02, 100.,0.02

* 4.2.2 Shoot
* RGUP     daily dry matter growth for shoot         kg.hm-2.d-1
* BIOMUP   total dry matter of shoot                 kg.hm-2

   RGUP  =GTW-RGRM
   BIOMUP=INTGRL(ISHOT,RGUP)

* 4.2.3 leaf, stem and boll
*  RWSM   daily drymatter growth for stem          kg.hm-2.d-1
*  RWLV   daily drymatter growth for leaf          kg.hm-2.d-1
*  RWBL   daily drymatter growth for boll          kg.hm-2.d-1
*  WLEAF  leaf weight                              kg.hm-2
*  WSTEM  stem weight                              kg.hm-2
*  BOLLW  total boll weight including abscision    kg.hm-2

   RWSM=AFGEN(FSMTB,PDT)*GROWTH
   FLV =AFGEN(FLVTB,PDT)
   RWLV=FLV*GROWTH
   RWLVA=RWLV-RLDR*WLEAF
   RWBL=AFGEN(FBLTB,PDT)*GROWTH

FUNCTION FLVTB= 0.0,0.,   2.5,0.65, 10.,0.7, 17.5,0.1,...
                40.0,0.05, 60.,0.0, 100.,0.
FUNCTION FSMTB= 0.0,0.,   2.5,0.35, 10.,0.3,  17.5,0.1,...
                40.0,0.05, 60.,0.0, 100.,0.
FUNCTION FBLTB= 0.0,0.,   2.5,0.00,  17.5,0.8,...
              40.0,0.9, 60.,1.0, 100.,1.0

   WLEAF=INTGRL(IWLV  ,RWLVA)
   WSTEM=INTGRL(ISTEM ,RWSM)
```

```
306
307
308   * Dry matter calculated by partitioning index method (measured)
309   * for check of LAI etc.
310     PIROOT=1.-PISHOT
311     PISHOT=-4E-05*PDT**2. + 0.0055*PDT + 0.7136
312
313     ASHOOT=TDRW*PISHOT
314   *  AROOT =TDRW*PIROOT
315      WSHOOT=WLEAF+WSTEM+BOLLW
316
317   * 5. Boll growth, development and abscission
318   * boll growth and abscission calculation by fixed boxcar train method
319
320   * 5.1 Boll number simulation
321   * BOLOUT     number of boll opening          number per plant
322   * BOLTOT     number of square,flower,boll    number per plant
323   * FALL1      abscision due to insufficient dry matter supply
324   * FALL2      abscision due to pest
325   * FALL3      abscision due to high temperature
326   * RTE        relative thermal effectiveness
327   * RTEBOL     RTE specially in cotton boll stage,
328   *            set base temperature to 10 oC
329   * STRBOL     ratio of supply and demand
330   * FALLBL     total  falled boll
331   * SQUARE     number of square                          number per plant
332   * FLOWER     number of flower                          number per plant
333   * SBOLL      number of small boll                      number per plant
334   * BBOLL      number of big boll                        number per plant
335   * GREENB     number of green boll (with certain yield)number per plant
336
337      GAMMA=GF/REAL(N)
338   *   G    =INTGRL(GI, DEVR)
339
340      INFL        =MAX(0.,RFNR)
341
342      CBOLL(1:N)  =MAX(0.,BOLL(1:N))/GAMMA
343
344      FLOW(1)     =INFL;                          ...
345      FLOW(2:33)  =CBOLL(1:32)*RTE;               ...
346      FLOW(34:N+1)=CBOLL(33:N)*RTEBOL
347
348      NETFLO(1:N) =FLOW(1:N)-FLOW(2:N+1)
349
350      RBOLL(1:5)  =NETFLO(1:5) -BOLL(1:5) *FALL2-BOLL(1:5) *FALL3;   ...
351      RBOLL(6:20) =NETFLO(6:20)-BOLL(6:20)*FALL1-BOLL(6:20)*         ...
352                                          FALL2-BOLL( 6:20)*FALL3;...
353      RBOLL(21:25)=NETFLO(21:25)-BOLL(21:25)*FALL2-BOLL(21:25)*FALL3;...
354      RBOLL(26:33)=NETFLO(26:33)-BOLL(26:33)*FALL1-BOLL(26:33)*      ...
355                                          FALL2-BOLL(26:33)*FALL3;...
356      RBOLL(34:N) =NETFLO(34:N) -BOLL(34:N) *FALL2-BOLL(34:N) *FALL3
357
358      OUTFL       =FLOW(N+1)
359
360      RTEBOL=AFGEN(RTEBTB,TAV)
361      FALL1 =AFGEN(FALLTB,STRBOL)
362      FALL2 =AFGEN(FAL2TB,PDT)
363      FALL3 =AFGEN(FAL3TB,TAV)
364   FUNCTION RTEBTB=-10.,0., 0.,0., 10.,0., 22.,1., 35.,1.
365   FUNCTION FALLTB=         0.,1., 1.,0.
366   FUNCTION FAL2TB=-10.,0., 0.,0.,2.5,0., 27.5,0.,60.,0.001, 80.,0.,100.,0.
```

```
FUNCTION FAL3TB=-10.,0., 0.,0., 30.,0., 45.,1.

   BOLL  =INTGRL(BOLLI ,RBOLL)
   BOLOUT=INTGRL(BOLL0I,OUTFL)
*   BOLTOT=ARSUMM(BOLL,1,N)

   SQUARE=ARSUMM(BOLL, 1,22)
   FLOWER=ARSUMM(BOLL,23,25)
   SBOLL =ARSUMM(BOLL,26,33)
   BBOLL =ARSUMM(BOLL,34,74)
   GREENB=ARSUMM(BOLL,60,74)

   FALLBL=FN-ARSUMM(BOLL,1,N)

* 5.2 Boll weight simulation
* WBOLL        boll weight of each box             g.plant-1
* WBOUT        boll weight of opening boll         g.plant-1
* WBTOT        total boll weight except open boll  g.plant-1
* WSOUT        boll weight per single open boll    g.boll-1
* RWBAGE       weight growth between each box      g.boll-1.
* WBSUM        total boll weight                   kg.hm-2
* STRESS       boll weight loss due to unenoght supply
* WSQU         square weight                       kg.hm-2
* WFLW         flower weight                       kg.hm-2
* WSBO         small boll weight                   kg.hm-2
* WBBO         big boll  weight                    kg.hm-2
* WGREEN       green boll weight                   kg.hm-2

PARAM BOLLW0=0.019

   WINFL        =INFL*BOLLW0

   CWBOLL(1:N)  =MAX(0.,WBOLL(1:N))/GAMMA

   WFLOW(1)     =WINFL;                      ...
   WFLOW(2:33)  =CWBOLL(1:32)*RTE;           ...
   WFLOW(34:N+1)=CWBOLL(33:N)*RTEBOL
   WNETFL(1:N)  =WFLOW(1:N)-WFLOW(2:N+1)

   WOUTFL       =WFLOW(N+1)

   RWBOL(1:5)   =WNETFL(1:5) + BOLL(1:5)*RWBAGE(1:5)-              ...
                 BOLL(1:5)*FALL2*WBAGE(1:5)-                       ...
                 BOLL(1:5)*FALL3*WBAGE(1:5);                      ...
   RWBOL(6:20)  =WNETFL(6:20)+ BOLL(6:20)*RWBAGE(6:20)-            ...
                 BOLL(6:20)*FALL1*WBAGE(6:20)-                     ...
                 BOLL(6:20)* FALL2*WBAGE(6:20)-                    ...
                 BOLL( 6:20)*FALL3*WBAGE(6:20);                    ...
   RWBOL(21:25) =WNETFL(21:25)+ BOLL(21:25)*RWBAGE(21:25)-         ...
                 BOLL(21:25)*FALL2*WBAGE(21:25)-                   ...
                 BOLL(21:25)*FALL3*WBAGE(21:25);                   ...
   RWBOL(26:33) =WNETFL(26:33)+ BOLL(26:33)*RWBAGE(26:33)-         ...
                 BOLL(26:33)*FALL1*WBAGE(26:33)-                   ...
                 BOLL(26:33)*FALL2*WBAGE(26:33)-                   ...
                 BOLL(26:33)*FALL3*WBAGE(26:33);                   ...
   RWBOL(34:N)  =WNETFL(34:N)+BOLL(34:N)*RWBAGE(34:N)*STRESS(34:N)- ...
                 BOLL(34:N) *FALL2*WBAGE(34:N)-                    ...
                 BOLL(34:N) *FALL3*WBAGE(34:N)

   WBOLL =INTGRL(BOLLI,RWBOL)
   WBOUT =INTGRL(ZERO,WOUTFL)
```

```
*   WBTOT =ARSUMM(WBOLL,1,N)

    WSOUT =WBOUT/NOTNUL(BOLOUT)

    WSQU  =ARSUMM(WBOLL,1,22)*PLANTS/1000.
    WFLO  =ARSUMM(WBOLL,23,25)*PLANTS/1000.
    WSBO  =ARSUMM(WBOLL,26,33)*PLANTS/1000.
    WBBO  =ARSUMM(WBOLL,34,74)*PLANTS/1000.
    WGREEN=ARSUMM(WBOLL,60,74)*PLANTS/1000.
    BOLLW=WSQU+WFLO+WSBO+WBBO+WBOUT*PLANTS/1000.


*5.3 Potential single boll weight
* AGE        boll age:  square,1-22, flower, 23-25, small boll, 26-33,
*                       big boll, 34-74, green boll, 60-74
* WBOLLP     potential single boll weight                      kg.hm-2
* WBMAX      maxium single boll weight, is a param of cultivar g
* WBAGE      single boll weight for each age(1-74)
* RWBT       daily demand for all boll growth              kg.hm-2.d-1
* WBP        total demand for all boll growth              kg.hm-2

  AGE(1)   = 1.;AGE(2) = 2.;AGE(3) = 3.;AGE(4) = 4.;AGE(5) = 5.;...
  AGE(6)   = 6.;AGE(7) = 7.;AGE(8) = 8.;AGE(9) = 9.;AGE(10)=10.;...
  AGE(11)  =11.;AGE(12)=12.;AGE(13)=13.;AGE(14)=14.;AGE(15)=15.;...
  AGE(16)  =16.;AGE(17)=17.;AGE(18)=18.;AGE(19)=19.;AGE(20)=20.;...
  AGE(21)  =21.;AGE(22)=22.;AGE(23)=23.;AGE(24)=24.;AGE(25)=25.;...
  AGE(26)  =26.;AGE(27)=27.;AGE(28)=28.;AGE(29)=29.;AGE(30)=30.;...
  AGE(31)  =31.;AGE(32)=32.;AGE(33)=33.;AGE(34)=34.;AGE(35)=35.;...
  AGE(36)  =36.;AGE(37)=37.;AGE(38)=38.;AGE(39)=39.;AGE(40)=40.;...
  AGE(41)  =41.;AGE(42)=42.;AGE(43)=43.;AGE(44)=44.;AGE(45)=45.;...
  AGE(46)  =46.;AGE(47)=47.;AGE(48)=48.;AGE(49)=49.;AGE(50)=50.;...
  AGE(51)  =51.;AGE(52)=52.;AGE(53)=53.;AGE(54)=54.;AGE(55)=55.;...
  AGE(56)  =56.;AGE(57)=57.;AGE(58)=58.;AGE(59)=59.;AGE(60)=60.;...
  AGE(61)  =61.;AGE(62)=62.;AGE(63)=63.;AGE(64)=64.;AGE(65)=65.;...
  AGE(66)  =66.;AGE(67)=67.;AGE(68)=68.;AGE(69)=69.;AGE(70)=70.;...
  AGE(71)  =71.;AGE(72)=72.;AGE(73)=73.;...
  AGE(74:N)=REAL(N)

PARAM RMBOLL=0.1143
PARAM WBMAX =7.

  WBAGE(1:N)=(WBMAX/(1.+219.*EXP(-RMBOLL*AGE(1:N))))
  RWBAGE(1)    =WBAGE(2)-WBAGE(1)        ;...
  RWBAGE(2:N-1)=(WBAGE(3:N)-WBAGE(2:N-1));...
  RWBAGE(N)    =0.
  RWB(1:N)     =RWBAGE(1:N)*MAX(0.,BOLL(1:N))*PLANTS/1000.

  RWBT  =ARSUMM(RWB,1,N)
*  WBP   =INTGRL(IWBOLL,RWBT)
*  WBOLLP=MAX(0.,BOLL)*WBAGE*PLANTS/1000.

  STRESS(1:33)=1.;...
  STRESS(34:N)=STRBOL


* Yield boll and weight including open boll, green boll
*  YIELDB=(GREENB+BOLOUT)*PLANTS
  YIELDW=(WGREEN+WBOUT) *PLANTS/1000.
*  OPENBL=BOLOUT*PLANTS
*  OPENBW=WBOUT *PLANTS/1000.
*  GREBLW=WGREEN*PLANTS/1000.
```

```
489     GREBL =GREENB*PLANTS
490
491   * 5.4 Boll shell weight
492   *     used for calculate seed cotton
493   * SHELL      weight of boll shell for boll in each age    g.boll-1
494
495     SHELL(1:25) =0.                                          ;...
496     SHELL(26:55)=2.3/(1.+21753.5329*EXP(-0.28475*AGE(26:55)));...
497     SHELL(56:N) =2.2195*EXP(-0.0038*AGE(56:N))
498
499   * 5.5 Seed cotton yield
500   *  SEDCOT    seed cotton in each boll              g.plant-1
501   *  COTOUT    seed cotton of open boll              kg.hm-2
502   *  COTT      seed cotton of total boll             kg.hm-2
503   *  COTGRE    seed cotton of total green boll       kg.hm-2
504   *  COTTON    seed cotton yield                     kg.hm-2
505   *  COTR      ratio of seedcotton and shell
506   *  RCOUT & RCGRE  are ratio COTR of openboll and greenboll,respectively
507
508      SEDCOT=WBOLL-SHELL*BOLL
509      COTOUT=(WBOUT-SHELL(N)*BOLOUT)*PLANTS/1000.
510   *   COTT  =ARSUMM(SEDCOT,1,N)*PLANTS/1000.
511      COTGRE=ARSUMM(SEDCOT,60,74)*PLANTS/1000.
512      COTTON=(COTGRE+COTOUT)
513      COTR  =MAX(0.,COTTON/NOTNUL(YIELDW))
514   *   RCOUT =MAX(0.,COTOUT/NOTNUL(OPENBW))
515   *   RCGRE =MAX(0.,COTGRE/NOTNUL(GREBLW))
516
517   * 5.6 Lint
518   * GOTOUT          lint percentage for open boll
519   * GOTGRN          lint percentage for green boll
520   * LINT            total lint                      kg.hm-2
521   * LINTO           lint of open boll               kg.hm-2
522   * LINTG           lint of green boll              kg.hm-2
523   * RLINT           average lint percentage
524   * LINTM           lint of open boll in chinese MU     kg.mu-1
525   * LINTT           total lint in chinese MU            kg.mu-1
526
527   PARAM GOTOUT=0.38; GOTGRN=0.34
528
529      LINTO=COTOUT*GOTOUT
530      LINTG=COTGRE*GOTGRN
531      LINT =LINTO+LINTG
532      RLINT=LINT/NOTNUL(COTOUT+COTGRE)
533      LINTM=LINTO/15.
534      LINTT=LINT /15.
535
536   * 5.7  Ratio of supply and demand
537   * STRBOL   supply/demand ratio, 1., no stress, 0., maximum stress
538
539     STRBL1=MIN (1., RWBL/NOTNUL(RWBT))
540     STRBOL=INSW(PDT-17.5, 1., MAX(0.,STRBL1) )
541
542   * 6.  Photosynthesis
543
544   * 6.1 Leaf CO2 assimulation
545   PARAM AMX=50.
546   FUNCTION AMPDTT=0.0,0.0, 2.5,1.0, 60.,1.0, 70.,0.1, 100.,0.
547       AMAX =AMX*AMPDT*AMTMP
548       AMPDT=AFGEN(AMPDTT,PDT)
549       AMTMP=MAX(0.0, 1.-0.003*(TAV-30.)**2.)
```

```
550
551   * 6.2 Daily gross CO2 assimulation
552   PARAM EFF =0.45
553   PARAM RHOS=0.2
554   PARAM KDF =1.0
555   PARAM SCP =0.2
556
557      CALL TOTASS (TIME, LAT , RDD, SCP,RHOS,AMAX, EFF, KDF, LAI,...
558                         DAYL, DTGA, FGROS,IABSD,BALNCE)
559
560
561   * 6.3 Carbohydrate production
562         GPHOT=DTGA*30./44.
563
564   * 6.4 Maintenance
565         MAINT=MINTS*MNTE*MNPDT
566         MINTS=WSQU *0.038 + WFLO*0.076 + WLEAF*0.0264 + ROOTA*0.038 + ...
567               WSTEM*0.006 + WSBO*0.038 + WBBO *0.032  + WBOUT*0.032
568         MNTE =2.**((TAV-28.0)/10.)
569         MNPDT=AFGEN(MNPDTT, PDT)
570   FUNCTION MNPDTT=0.0,0.0, 2.5,1.0, 27.5,0.9, 60.,0.5, 70.,0., 100., 0.
571
572   * 6.5 Growth
573         GTW  =GPHOT - MAINT - GPHOT*(1.-1./ASRQ)
574         ASRQ =AFGEN(ASRQTB, PDT)
575   FUNCTION ASRQTB=0.,1.42, 2.5,1.42, 17.5,1.42, 27.5,1.64, 100.,1.64
576
577   * 6.6 Dry matter
578         TDRW =INTGRL(IW, GTW)
579
580   TRANSLATION_GENERAL DRIVER='EUDRIV'
581   END
582
583   PARAM FILM=1.
584   END
585
586   * CRI37, early matured cotton
587   PARAM FILM=0.;PLANTS=90000.
588   PARAM VI=0.98;
589   PARAM FFB=5.
590   PARAM GOTOUT=0.35;GOTGRN=0.32
591   END
592
593   * non-management
594   PARAM FILM=0.
595   PARAM VI=0.82;PLANTS=60000.
596   PARAM FFB=9.
597   PARAM GOTOUT=0.34;GOTGRN=0.31
598   END
599
600
601   STOP
602
603   * 7. SUBROUTINES
604
605   *---------------------------------------------------------------------*
606   *  SUBROUTINE ASTRO                                                   *
607   *  Purpose: This subroutine calculates astronomic daylength,          *
608   *           diurnal radiation characteristics such as the daily       *
609   *           integral of sine of solar elevation and solar constant.   *
610   *                                                                     *
```

```
611   *  FORMAL PARAMETERS:  (I=input,O=output,C=control,IN=init,T=time)     *
612   *  name   type meaning                                      units  class *
613   *  ----   ---- -------                                      -----  ----- *
614   *  DOY     R4  Daynumber (Jan 1st = 1)                        -      I  *
615   *  LAT     R4  Latitude of the site                        degrees   I  *
616   *  SC      R4  Solar constant                              J m-2 s-1 O  *
617   *  SINLD   R4  Seasonal offset of sine of solar height        -      O  *
618   *  COSLD   R4  Amplitude of sine of solar height              -      O  *
619   *  DAYL    R4  Astronomic daylength (base = 0 degrees)        h      O  *
620   *  DSINBE  R4  Daily total of effective solar height          s      O  *
621   *                                                                       *
622   *-----------------------------------------------------------------------*
623         SUBROUTINE ASTRO (DOY, LAT, SC , SINLD, COSLD, DAYL, DSINBE)
624         IMPLICIT REAL (A-Z)
625
626   *-----PI and conversion factor from degrees to radians
627         PI    = 3.141592654
628         RAD   = PI/180.
629
630   *-----declination of the sun as function of daynumber (DOY)
631         DEC   = -ASIN (SIN (23.45*RAD)*COS (2.*PI*(DOY+10.)/365.))
632
633   *-----SINLD, COSLD and AOB are intermediate variables
634         SINLD = SIN (RAD*LAT)*SIN (DEC)
635         COSLD = COS (RAD*LAT)*COS (DEC)
636         AOB   = SINLD/COSLD
637
638   *-----daylength (DAYL)
639         DAYL   = 12.0*(1.+2.*ASIN (AOB)/PI)
640         DSINBE = 3600.*(DAYL*(SINLD+0.4*(SINLD*SINLD+COSLD*COSLD*0.5))+
641        &         12.0*COSLD*(2.0+3.0*0.4*SINLD)*SQRT (1.-AOB*AOB)/PI)
642
643   *-----solar constant (SC) and daily extraterrestrial radiation
644         SC  = 1370.*(1.+0.033*COS (2.*PI*DOY/365.))
645
646         RETURN
647         END
648
649   *-----------------------------------------------------------------------*
650   *  SUBROUTINE TOTASS                                                    *
651   *  Purpose: This subroutine calculates daily total gross                *
652   *           assimilation (DTGA) by performing a Gaussian integration    *
653   *           over time. At three different times of the day,             *
654   *           radiation is computed and used to determine assimilation    *
655   *           whereafter integration takes place                         *
656   *                                                                       *
657   *  FORMAL PARAMETERS:  (I=input,O=output,C=control,IN=init,T=time)      *
658   *  name   type meaning                                      units  class *
659   *  ----   ---- -------                                      -----  ----- *
660   *  DOY     R4  Daynumber (January 1 = 1)                      -      I  *
661   *  LAT     R4  Latitude of the site                        degrees   I  *
662   *  DTR     R4  Daily total of global radiation             J/m2/d    I  *
663   *  SCP     R4  Scattering coefficient of leaves for visible           *
664   *              radiation (PAR)                                -      I  *
665   *  RHOS
666   *  AMAX    R4  Assimilation rate at light saturation       kg CO2/   I  *
667   *                                                         ha leaf/h     *
668   *  EFF     R4  Initial light use efficiency               kg CO2/J/  I  *
669   *                                                         ha/h m2 s     *
670   *  KDF     R4  Extinction coefficient for diffuse light               I  *
671   *  LAI     R4  Leaf area index                             ha/ha     I  *
```

```
*  DAYL    R4  Astronomic daylength (base = 0 degrees)       h      O  *
*  DTGA    R4  Daily total gross assimilation          kg CO2/ha/d O  *
*  FGROS
*  IABSD
*  BALNCE
*
*  SUBROUTINES and FUNCTIONS called : ASTRO, ASSIM                    *
*-----------------------------------------------------------------------*
       SUBROUTINE TOTASS (DOY,  LAT , DTR, SCP,RHOS,AMAX, EFF, KDF, LAI,
     &                   DAYL, DTGA,FGROS,IABSD,BALNCE)
       IMPLICIT REAL(A-Z)
       REAL XGAUSS(3), WGAUSS(3)
       INTEGER I1, IGAUSS

       DATA IGAUSS /3/
       DATA XGAUSS /0.112702, 0.500000, 0.887298/
       DATA WGAUSS /0.277778, 0.444444, 0.277778/

       PI   = 3.141592654

       CALL ASTRO(DOY,LAT,SC,SINLD,COSLD,DAYL,DSINBE)

*---assimilation set to zero and three different times of the day (HOUR)
       DTGA = 0.
       IABSD= 0.
       DO 10 I1=1,IGAUSS

*-------at the specified HOUR, radiation is computed and used to compute
*       assimilation
        HOUR = 12.0+DAYL*0.5*XGAUSS(I1)

*-------sine of solar elevation
        SINB  = MAX (0., SINLD+COSLD*COS (2.*PI*(HOUR+12.)/24.))

*-------diffuse light fraction (FRDF) from atmospheric
*       transmission (ATMTR)
        PAR   = 0.5*DTR*SINB*(1.+0.4*SINB)/DSINBE
        ATMTR = PAR/(0.5*SC*SINB)

        IF (ATMTR.LE.0.22) THEN
            FRDF = 1.
        ELSE IF (ATMTR.GT.0.22 .AND. ATMTR.LE.0.35) THEN
            FRDF = 1.-6.4*(ATMTR-0.22)**2
        ELSE
            FRDF = 1.47-1.66*ATMTR
        END IF

        FRDF = MAX (FRDF, 0.15+0.85*(1.-EXP (-0.1/SINB)))

*-------diffuse PAR (PARDF) and direct PAR (PARDR)
        PARDF = PAR * FRDF
        PARDR = PAR - PARDF

        CALL ASSIMS (SCP,RHOS,AMAX,EFF,KDF,LAI,SINB,PARDR,PARDF,
     &               FGROS, IABS,BALNCE)

*-------integration of assimilation rate to a daily total (DTGA)
        IABSD=IABSD +  IABS*WGAUSS(I1)
        DTGA = DTGA + FGROS*WGAUSS(I1)

10    CONTINUE
```

```
      DTGA = DTGA * DAYL

      RETURN
      END


*-----------------------------------------------------------------------*
*  SUBROUTINE ASSIMS                                                    *
*  Purpose: This subroutine performs a Gaussian integration over        *
*           depth of canopy by selecting three different LAI's and      *
*           computing assimilation at these LAI levels. The             *
*           integrated variable is FGROS.                               *
*           # Soil refection is included                                *
*           # with 5-point Gausssian integration                        *
*                                                                       *
*  FORMAL PARAMETERS:  (I=input,O=output,C=control,IN=init,T=time)      *
*  name   type meaning                                    units  class  *
*  ----   ---- -------                                    -----  -----  *
*  SCP     R4  Scattering coefficient of leaves for visible            *
*              radiation (PAR)                              -       I   *
*  RHOS
*  AMAX    R4  Assimilation rate at light saturation      kg CO2/    I   *
*                                                        ha leaf/h      *
*  EFF     R4  Initial light use efficiency              kg CO2/J/  I   *
*                                                        ha/h m2 s      *
*  KDF     R4  Extinction coefficient for leaves                    I   *
*  LAI     R4  Leaf area index                             ha/ha    I   *
*  SINB    R4  Sine of solar height                          -      I   *
*  PARDR   R4  Instantaneous flux of direct radiation (PAR) W/m2    I   *
*  PARDF   R4  Instantaneous flux of diffuse radiation(PAR) W/m2    I   *
*  FGROS   R4  Instantaneous assimilation rate of         kg CO2/   O   *
*              whole canopy                              ha soil/h      *
*  IABS
*  BALNCE
*                                                                       *
*-----------------------------------------------------------------------*
      SUBROUTINE ASSIMS (SCP,RHOS,AMAX,EFF,KDF,LAI,SINB,
     $                   PARDR,PARDF,   FGROS, IABS,BALNCE)
      IMPLICIT REAL(A-Z)
      REAL XGAUSS(5), WGAUSS(5)
      INTEGER I1, I2, IGAUSS

*-----Gauss weights for three point Gauss
      DATA IGAUSS /5/
      DATA XGAUSS /0.0469101,0.2307534,0.5,      0.7692465,0.9530899/
      DATA WGAUSS /0.1184635,0.2393144,0.2844444,0.2393144,0.1184635/

*-----reflection of horizontal and spherical leaf angle distribution
      SQV  = SQRT(1.-SCP)
      REFH = (1.-SQV)/(1.+SQV)
      REFS = REFH*2./(1.+1.6*SINB)
      ETA1 = (REFH-RHOS)/(RHOS-1./REFH)
      ETA2 = (REFS-RHOS)/(RHOS-1./REFS)

*-----extinction coefficient for direct radiation and total direct flux
      CLUSTF = KDF / (0.8*SQV)
      KBL    = (0.5/SINB) * CLUSTF
      KDRT   = KBL * SQV
      CORRV1 = ETA1*EXP(-2.*KDF *LAI)
      CORRV2 = ETA2*EXP(-2.*KDRT*LAI)
```

```
794         DENOM1 = 1.+CORRV1
795         DENOM2 = 1.+CORRV2
796
797   *-----selection of depth of canopy, canopy assimilation is set to zero
798         FGROS = 0.
799         IABS  = 0.
800
801         DO 10 I1=1,IGAUSS
802            LAIC = LAI * XGAUSS(I1)
803
804   *--------absorbed fluxes per unit leaf area: diffuse flux, total direct
805   *        flux, direct component of direct flux.
806            VISDF = (1.-REFH)*PARDF*KDF  *
807        &            (EXP(-KDF*LAIC) + EXP(KDF*LAIC)*CORRV1/REFH)/DENOM1
808            VIST  = (1.-REFS)*PARDR*KDRT *
809        &            (EXP(-KDRT*LAIC) + EXP(KDRT*LAIC)*CORRV2/REFS)/DENOM2
810            VISD  = (1.-SCP) *PARDR*KBL  *EXP (-KBL  *LAIC)
811
812   *--------absorbed flux (J/M2 leaf/s) for shaded leaves and assimilation of
813   *        shaded leaves
814            VISSHD = VISDF + VIST - VISD
815            IF (AMAX.GT.0.) THEN
816               FGRSH  = AMAX * (1.-EXP(-VISSHD*EFF/AMAX))
817            ELSE
818               FGRSH = 0.
819            END IF
820
821   *--------direct flux absorbed by leaves perpendicular on direct beam and
822   *        assimilation of sunlit leaf area
823
824            VISPP  = (1.-SCP) * PARDR / SINB
825            FGRSUN = 0.
826            IABSUN = 0.
827            DO 20 I2=1,IGAUSS
828               VISSUN = VISSHD + VISPP * XGAUSS(I2)
829               IF (AMAX.GT.0.) THEN
830                  FGRS = AMAX * (1.-EXP(-VISSUN*EFF/AMAX))
831               ELSE
832                  FGRS = 0.
833               END IF
834               FGRSUN = FGRSUN + FGRS   * WGAUSS(I2)
835               IABSUN = IABSUN + VISSUN * WGAUSS(I2)
836   20       CONTINUE
837
838   *--------fraction sunlit leaf area (FSLLA) and local assimilation
839   *        rate (FGL)
840            FSLLA = CLUSTF * EXP(-KBL*LAIC)
841            FGL   = FSLLA  * FGRSUN + (1.-FSLLA) * FGRSH
842            IABSL = FSLLA  * IABSUN + (1.-FSLLA) * VISSHD
843
844   *--------integration of local assimilation rate to canopy
845   *        assimilation (FGROS)
846            FGROS = FGROS + FGL   * WGAUSS(I1)
847            IABS  = IABS  + IABSL * WGAUSS(I1)
848
849   10    CONTINUE
850         FGROS = FGROS * LAI
851         IABS  = IABS  * LAI
852
853   *     Warning:      in the expression for ISOIL
854   *     the use of REFS and REFH is not a mistake
```

```
855         IREFL  = PARDR*(REFS + CORRV2/REFS)/DENOM2 +
856        &         PARDF*(REFH + CORRV1/REFH)/DENOM1
857         ISOIL  = (1.-REFS)*PARDR*(EXP(-KDRT*LAI)-EXP(KDRT*LAI)
858        &                         *CORRV2/REFS)/DENOM2 +
859        &         (1.-REFH)*PARDF*(EXP(-KDF*LAI)-EXP(KDF*LAI)
860        &                         *CORRV1/REFH)/DENOM1
861         ITOT   = PARDF + PARDR
862         BALNCE = 100. * (ITOT-IREFL-ISOIL-IABS)/ITOT
863
864         RETURN
865         END
866
867   ENDJOB
868
```